Longitudinal Dynamics CAS 2011 - Chios

```
(%i2)  E0 : 0.93826E9 $
       c  : 2.9979e8 $
       rho_PS :  70 $
       R_PS   : 100 $
       R_PSB  :  25 $
       h_PSB  :   1 $
       h_PS   :   8 $
       a_PSB  : 0.0617 $
       a_PS   : 0.027  $
       V_RF_PSB : 8000 $
       P_ext  : 26e9  $
```

1) Calculate the magnetic field in the PS at injection and ejection.

```
(%i13) E_kin_inj : 1.4E9$
```

```
(%i14) E_tot_inj : E_kin_inj + E0 $
       float(%) *1e-9*'GeV;
```

(%o15) $2.338\, GeV$

```
(%i16) y_inj : E_tot_inj / E0 ;
```

(%o16) 2.492

```
(%i17) b_inj : sqrt(1-1/ y_inj ^2);
```

(%o17) 0.916

```
(%i18) P_inj : b_inj * E_tot_inj $
       float(%) *1e-9*'GeV;
```

(%o19) $2.142\, GeV$

Field in Tesla

```
(%i20) B_inj : P_inj *1e-9 / (0.3 * rho_PS);
```

(%o20) 0.102

```
(%i21) B_ext : P_ext *1e-9 / (0.3 * rho_PS);
```

(%o21) 1.238

2) At injection, calculate and compare the RF frequency in the PS and in the PSB.

```
(%i22) v_RF_PS_inj : b_inj * c * h_PS / (2 * %pi * R_PS) $
       float(% *1e-6*'MHz);
```

(%o23) $3.496\,MHz$

```
(%i24) v_RF_PSB : b_inj * c * h_PSB / (2 * %pi * R_PSB) $
       float(% *1e-6*'MHz);
```

(%o25) $1.748\,MHz$

3) What is the synchronous phase in the PS at injection?

No acceleration, before transition -> phi_s = 0

4) Assuming that the orbit remains the same during the acceleration, how does the RF frequency change between injection and ejection in the PS?

```
(%i26) E_tot_ext : sqrt( E0^2 + P_ext^2 ) $ float(%) *1e-9*'GeV;
       y_ext : E_tot_ext / E0;
       b_ext : sqrt(1-1/ y_ext ^2);
```

(%o27) $26.02\,GeV$

(%o28) 27.73

(%o29) .9993

```
(%i30) v_RF_ext : v_RF_PS_inj * b_ext / b_inj $
       float(% *1e-6*'MHz);
```

(%o31) $3.815\,MHz$

5) Should a phase jump system be implemented in the PS?

```
(%i32) y_tr : 1/sqrt(a_PS);
```

(%o32) 6.086

Since this is between y_inj and y_ext => transition is crossed.

6) At PS injection, the maximum relative momentum spread and the longitudinal emittance are Δp/p = ± 2.10-3 and $\epsilon = \pi \Delta E \Delta t = 0.9$ eV.s. Assuming a perfect elliptic area, calculate the value of the bunch length t_b. Compare your result to the value given in the introduction.

```
(%i33) 'dp/p = 1/%beta^2 * 'dE/E;
```

(%o33) $\frac{dp}{p} = \frac{dE}{\beta^2\,E}$

```
(%i34) rel_delta_P_inj : 2e-3 $
       delta_E_inj : b_inj^2 * E_tot_inj * rel_delta_P_inj $
       float(% *1e-6*'MeV);
```

(%o36) $3.924\, MeV$

```
(%i37) epsilon : 0.9;
```

(%o37) 0.9

```
(%i38) solve(%epsilon = %pi * 'delta_E_inj * t_b/2, t_b);
```

(%o38) $[t\_b = \frac{2\,\epsilon}{\pi\, delta\_E\_inj}]$

```
(%i39) t_bunch : 2 * epsilon / (%pi * delta_E_inj) $
       float(% * 1e9*'ns);
```

(%o40) $146.0\, ns$

7) What is the value of VRF at injection in the PS for a perfect longitudinal matching?

use Delta E_b = Delta E_RF * sin(phi_max/2)

```
(%i41) phi_max_PSB : 2*%pi * v_RF_PSB * t_bunch/2 $ float(%);
```

(%o42) 0.802

```
(%i43) w(x):=sin(x/2) $
       w2(x):=sqrt(cos(phim/2)^2-cos(x/2)^2);
```

(%o44) $\mathrm{w2}\left(x\right) := \sqrt{\cos\left(\frac{phim}{2}\right)^2 - \cos\left(\frac{x}{2}\right)^2}$

```
(%i45) phim : %pi - phi_max_PSB $
       wxplot2d([w(x),w2(x)],[x,0,2*%pi],[xlabel,phi],[ylabel,'W_PSB],
       [legend,false],[color, blue]);
```

plot2d: expression evaluates to non-numeric value somewhere in plotting range.

(%t46) 

(%t46) (%o46)

```
(%i47) phi_max_PS : 2*%pi * v_RF_PS_inj * t_bunch/2 $ float(%);
```

(%o48) 1.604

```
(%i49) phim : phi_max_PS $
       w(x):=cos(x/2) $
       w2(x):=sqrt(sin(phim/2)^2-sin(x/2)^2)$
       wxplot2d([w(x),w2(x)],[x,-%pi,%pi],[xlabel,phi],[ylabel,'W_PS],
       [legend,false],[color, blue]);
```

plot2d: expression evaluates to non-numeric value somewhere in plotting range.

(%t52) 

(%o52)

The phase extension is twice bigger at the frequency of the PS. The energy spread for the bunch stays the same.

```
(%i53) eta_PS  : 1/y_inj^2-a_PS;
       eta_PSB : 1/y_inj^2-a_PSB;
```

(%o53) 0.134

(%o54) .09931

```
(%i55) delta_E_inj *1e-6 *'MeV;
```

(%o55) $3.924\, MeV$

```
(%i56) delta_E_RF_PS : 1/sin(phi_max_PS/2) * delta_E_inj $ float(%)*1e-6 *'MeV;
```

(%o57) $5.459\, MeV$

```
(%i58) delta_E_RF_PSB : b_inj * sqrt(V_RF_PSB*E_tot_inj*2/(%pi * eta_PSB * h_PSB))$
       float(% *1e-6 *'MeV);
```

(%o59) $10.03\, MeV$

```
(%i60) V_RF_PS : (delta_E_RF_PS/b_inj)^2/E_tot_inj/2 * %pi* eta_PS * h_PS $
       float(% *1e-3*'kV);
```

(%o61) $25.58\, kV$

8) From the peak RF voltage found in question 7, calculate the PS synchrotron frequency f_s and the synchrotron tune Q_s at injection. Does it verify Q_s << 1 ?

```
(%i62) f_s : c/(2*%pi*R_PS)*sqrt(V_RF_PS*eta_PS*h_PS/(2*%pi*E_tot_inj)) $
       float(% * 'Hz);
```

(%o63) $651.9\, Hz$

```
(%i64) Q_s : f_s/(v_RF_PS_inj/h_PS) $
       float(%);
```

(%o65) .001492